```
      ID = KK + J
      IF (J .GT. K) ID = JJ + K
      DIST = D(ID)
      IF (DIST .LT. U(K)) U(K) = DIST
      IF (A .LE. U(K)) GOTO 2
      A = U(K)
      NEXT = K
    2 CONTINUE
      J = NEXT
      P(I) = NEXT
      W(I - 1) = U(NEXT)
      IND(J) = 1
    3 CONTINUE
C
C        COMPUTE THE ULTRAMETRIC DISTANCES BY FINDING THE MAXIMUM OF
C        EACH ADJACENT PAIR OF PREVIOUSLY COMPUTED DISTANCES IN W( ).
C
      N1 = N - 1
      DO 5 I = 1, N1
      N2 = N - I
      DO 4 J = 1, N2
      K = J + I
      L = MAX0(P(J), P(K))
      M = MIN0(P(J), P(K))
      IU = ((L - 1) * (L - 2)) / 2 + M
      U(IU) = W(J)
      IF (J .EQ. N2) GOTO 4
      IF (W(J + 1) .GT. W(J)) W(J) = W(J + 1)
    4 CONTINUE
    5 CONTINUE
      RETURN
      END
```

## Algorithm AS 103

# Psi (Digamma) Function

By J. M. Bernardo

*University College London, Britain*

*Keywords*: PSI; DIGAMMA; BETA DENSITIES; GAMMA DENSITIES

### Language

ISO Fortran

### Description and Purpose

A routine is presented to compute

$$\psi(x) = d\{\log \Gamma(x)\}/dx = \Gamma'(x)/\Gamma(x)$$

the psi or digamma function, for real positive values of $x$.

While the functions $\Gamma(x)$ and $\log \Gamma(x)$ are provided in most systems, this is not so with $\psi(x)$ which nevertheless often occurs in statistical practice, particularly when Beta or Gamma densities are involved.

## Method

For real positive $x$, $\psi(x)$ is a concave increasing function of $x$ which satisfies the following relations (Abramowitz and Stegun, 1964, pp. 258–259):

$$\psi(1) = -\gamma \simeq -0.5772156649, \tag{1}$$

$$\psi(1+x) = \psi(x) + \frac{1}{x}, \tag{2}$$

$$\psi(1+x) = -\gamma + \sum_{n=1}^{\infty} \frac{x}{n(n+x)} \quad (x \neq -1, -2, -3, \ldots), \tag{3}$$

$$\psi(x) = \log x - \frac{1}{2x} - \frac{1}{12x^2} + \frac{1}{120x^4} - \frac{1}{252x^6} + O\left(\frac{1}{x^8}\right) \quad (x \to \infty). \tag{4}$$

Moreover, using (1), (2) and (3)

$$\psi(x) = -\gamma - \frac{1}{x} + O(x) \quad (x \to 0). \tag{5}$$

The routine is presented in *FUNCTION* form with two parameters. The first parameter, called *X*, is the independent variable. A second parameter called *IFAULT* is set to 0 if *X* is positive and to 1 otherwise. In the latter case, the function *DIGAMA* is set to the (arbitrary) value 0.

*DIGAMA* is computed from (5) if $0 < x \leqslant S$ and from (4) if $x \geqslant C$. The values of the constants $S$ and $C$ that we have chosen are $S = 1{\cdot}0E-5$ and $C = 8{\cdot}5$. This makes the relative error committed by truncation smaller than $1{\cdot}0E-10$ in both cases.

If $S < x < C$, equation (2) is repeatedly used to express $\psi(x)$ in terms of $\psi(x+n)$ with $(x+n) \geqslant C$. The value of $\psi(x+n)$ is then computed, as before, using the Stirling expansion (4). This is remarkably more efficient than the direct computation of $\psi(x)$ from one of its series expansions.

## Structure

*FUNCTION DIGAMA (X, IFAULT)*

*Formal parameters*

| | | |
|---|---|---|
| *X* | real | input: parameter of the function |
| *IFAULT* | integer | output: fault indicator |

*Failure indications*

*IFAULT* = 0 if the parameter is positive
*IFAULT* = 1 otherwise

## Time and Accuracy

Both accuracy and consumed time will depend on the computer used. In an IBM 370/158, five correct digits are obtained with an average consumed time per calculated value of $3{\cdot}1 \times 10^{-3}$ sec.

## Acknowledgements

I am indebted to Mr D. Walley, to the editor and to one referee for many helpful comments.

## Reference

Abramowitz, M. and Stegun, I. A. (eds) (1964). *Handbook of Mathematical Functions*. New York: Dover.

```
      FUNCTION DIGAMA(X, IFAULT)
C
C        ALGORITHM AS 103  APPL. STATIST. (1976) VOL.25, NO.3
C
C        CALCULATES DIGAMMA(X) = D(LOG(GAMMA(X))) / DX
C
C        SET CONSTANTS. SN= NTH STIRLING COEFFICIENT, D1=DIGAMMA(1.0)
C
      DATA S, C, S3, S4, S5, D1 /1.0E-5, 8.5, 8.333333333E-2,
     * 8.333333333E-3, 3.968253968E-3, -0.5772156649/
C
C        CHECK ARGUMENT IS POSITIVE
C
      DIGAMA = 0.0
      Y = X
      IFAULT = 1
      IF (Y .LE. 0.0) RETURN
      IFAULT = 0
C
C        USE APPROXIMATION IF ARGUMENT .LE. S
C
      IF (Y .GT. S) GOTO 1
      DIGAMA = D1 - 1.0 / Y
      RETURN
C
C        REDUCE TO DIGAMA(X+N), (X+N) .GE. C
C
    1 IF (Y .GE. C) GOTO 2
      DIGAMA = DIGAMA - 1.0 / Y
      Y = Y + 1.0
      GOTO 1
C
C        USE STIRLING IF ARGUMENT .GE. C
C
    2 R = 1.0 / Y
      DIGAMA = DIGAMA + ALOG(Y) - 0.5 * R
      R = R * R
      DIGAMA = DIGAMA - R * (S3 - R * (S4 - R * S5))
      RETURN
      END
```

## Algorithm AS 104

# BLUS Residuals

By R. W. Farebrother

*University of Manchester, Britain*

*Keywords*: THEIL'S BLUS RESIDUALS

### Language

Algol 60

### Description and Purpose

This paper is concerned with the estimation of the $n \times 1$ matrix of disturbances $\boldsymbol{\varepsilon}$ in the standard linear model

$$\mathbf{y} = \mathbf{X}\boldsymbol{\beta} + \boldsymbol{\varepsilon}, \quad E\boldsymbol{\varepsilon} = \mathbf{0}, \quad E\boldsymbol{\varepsilon}\boldsymbol{\varepsilon}' = \sigma^2 \mathbf{I}_n,$$

where $\mathbf{y}$ is an $n \times 1$ matrix of observations on a random variable, $\mathbf{X}$ is an $n \times k$ full column